Heat Equation Problem -Mixed BCs - 11-09-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[10]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Previous version - *Heat Equation Problem - Mixed BCs 10-16-16.* Much earlier work circa 2008.

Purpose

I continue to solve heat equation problems from Chapter 3 of *Numerical and Analytical Methods for Scientists and Engineers, Using Mathematica*, Daniel Dubin. The example (p 221) considers a heat equation with homogeneous mixed boundary conditions.

I construct an Association that encapsulates information about this problem. I then apply the function *DSolveHeatEquation* that attempts to solve this problem directly using the Mathematica function *DSolve*. To illustrate the solution of a complete problem, I construct a possible initial condition consistent with the boundary conditions (see Appendix).

In[12]:=

```
A1 =
  Module[{description, pde, bcL, bcR, ic, eqns,
    assumptions, substitutions, simplifications, names, values},
   description = "Dubin example, p221 Homogeneous heat equation
      with homogeneous mixed boundary conditions";
   pde = D[T[x, t], t] - χ D[T[x, t], {x, 2}] == 0;
   bcL = κ Derivative[1, 0][T][0, t] - a T[0, t] == 0;
   bcR = κ Derivative[1, 0][T][L, t] + b T[L, t] == 0;
   ic = T[x, 0] == 1 + (a x)/κ - (x^2 (a b L + a κ + b κ))/(L κ (b L + 2 κ));
   eqns = {pde, bcL, bcR, ic};
   assumptions = {L > 0, χ > 0};
   substitutions = {K[1] → n};
   simplifications = {n ∈ Integers};
   values = {description, pde, bcL, bcR,
     ic, eqns, assumptions, substitutions, simplifications};
   names = {"description", "pde", "bcL", "bcR", "ic", "eqns",
     "assumptions", "substitutions", "simplifications"};
   AssociationThread[names, values]];

Module[{soln, G},
 soln = DSolveHeatEquation[A1];
 AppendTo[A1, "soln" → soln];
 Print@ShowPDESetup[A1];
 A1["soln"]]
```

Dubin example, p221 Homogeneous heat equation with homogeneous mixed boundary conditions



Out[13]=

```
T^(0,1)[x, t] == χ T^(2,0)[x, t]
```

DSolve cannot immediately solve this problem.

Use separation of variables. Assume T(x,t) = $\mathcal{T}$(t)ψ(x)

In[14]:=

```
w[1] = A1["pde"] /. T → Function[{x, t}, 𝒯[t] ψ[x]];
w[1] = MapEqn[(# / (𝒯[t] ψ[x])) &, w[1]] // Expand
```

Out[15]=

$$\frac{\mathcal{T}'[t]}{\mathcal{T}[t]} - \frac{\chi\,\psi''[x]}{\psi[x]} == 0$$

The separated equations are

In[16]:=

```
w[2] = {w[1][[1, 1]] == -λ, w[1][[1, 2]] == λ}
```

Out[16]=

$$\left\{\frac{\mathcal{T}'[t]}{\mathcal{T}[t]} == -\lambda,\ -\frac{\chi\,\psi''[x]}{\psi[x]} == \lambda\right\}$$

The latter constitutes a Sturm-Liouville equation problem that DSolve can handle.

In[17]:=

```
w[3] = DSolve[{χ ψ''[x] == -λ ψ[x], κ ψ'[0] == a ψ[0], κ ψ'[L] == -b ψ[L]},
    ψ[x], x, Assumptions → {χ ∈ Reals}][[1, 1]] /. λ → λn
```

Out[17]=

$$\psi[x] \to \begin{cases} C[1]\left(\mathrm{Sin}\left[\frac{x\sqrt{\lambda_n}}{\sqrt{\chi}}\right] + \left(\kappa\,\mathrm{Cos}\left[\frac{x\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}\right)\Big/\left(a\sqrt{\chi}\right)\right) & -a\,b\,\mathrm{Sin}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right] - \left(a\,\kappa\,\mathrm{Cos}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}\right)\Big/\left(\sqrt{\chi}\right) - \left(b\,\kappa\,\mathrm{Cos}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}\right)\Big/\left(\sqrt{\chi}\right) + \frac{1}{\chi}\kappa^2\,\mathrm{Sin}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\lambda_n == 0\ \&\&\ \chi \in \mathrm{Reals} \\ 0 & \mathrm{True} \end{cases}$$

The eigenvalues satisfy

In[18]:=

$$w[4] = -a\,b\,\mathrm{Sin}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right] - \frac{a\,\kappa\,\mathrm{Cos}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}}{\sqrt{\chi}} - \frac{b\,\kappa\,\mathrm{Cos}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}}{\sqrt{\chi}} + \frac{\kappa^2\,\mathrm{Sin}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\lambda_n}{\chi} == 0$$

Out[18]=

$$-a\,b\,\mathrm{Sin}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right] - \frac{a\,\kappa\,\mathrm{Cos}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}}{\sqrt{\chi}} - \frac{b\,\kappa\,\mathrm{Cos}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}}{\sqrt{\chi}} + \frac{\kappa^2\,\mathrm{Sin}\left[\frac{L\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\lambda_n}{\chi} == 0$$

and the eigenfunctions are

In[19]:=

$$w[5] = \psi_n[x] \to \left(\mathrm{Sin}\left[\frac{x\sqrt{\lambda_n}}{\sqrt{\chi}}\right] + \frac{\kappa\,\mathrm{Cos}\left[\frac{x\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}}{a\sqrt{\chi}}\right)$$

Out[19]=

$$\psi_n[x] \to \mathrm{Sin}\left[\frac{x\sqrt{\lambda_n}}{\sqrt{\chi}}\right] + \frac{\kappa\,\mathrm{Cos}\left[\frac{x\sqrt{\lambda_n}}{\sqrt{\chi}}\right]\sqrt{\lambda_n}}{a\sqrt{\chi}}$$

The time dependence is

In[20]:=

```
w[6] = DSolve[w[2][[1]], 𝒯[t], t][[1, 1]] /. λ → λn /. C[1] → 1 /. 𝒯 → 𝒯n
```

Out[20]=

$$\mathcal{T}_n[t] \to e^{-t\,\lambda_n}$$

The solution of the pde is a sum over these eigenfunctions

In[21]:=
```
w[7] = T[x, t] == Sum[A_n 𝒯_n[t] ψ_n[x], {n, 1, ∞}]
```
$$w[7] = T[x, t] == \sum_{n=1}^{\infty} A_n \mathcal{T}_n[t] \psi_n[x]$$

Out[21]=
$$T[x, t] == \sum_{n=1}^{\infty} A_n \mathcal{T}_n[t] \psi_n[x]$$

I follow Dubin and solve the eigenvalue problem for a simplified set of parameters

In[22]:=
$$def[\Lambda] = \Lambda == \frac{L\sqrt{\lambda_n}}{\sqrt{\chi}};$$
```
def[b] = b == a;
def[a] = a == 2 κ / L;
```

In[26]:=
```
w[8] = w[4] /. Sol[def[Λ], λn] /. Sol[def[b], b] /. Sol[def[a], a] // PowerExpand
```

Out[26]=
$$-\frac{4\kappa^2 \Lambda \mathrm{Cos}[\Lambda]}{L^2} - \frac{4\kappa^2 \mathrm{Sin}[\Lambda]}{L^2} + \frac{\kappa^2 \Lambda^2 \mathrm{Sin}[\Lambda]}{L^2} == 0$$

In[27]:=
```
w[9] = MapEqn[(# L^2 / κ^2) &, w[8]] // Expand
```

Out[27]=
$$-4\Lambda \mathrm{Cos}[\Lambda] - 4\mathrm{Sin}[\Lambda] + \Lambda^2 \mathrm{Sin}[\Lambda] == 0$$

In[28]:=
```
Clear[FcnΛ];
FcnΛ[Λ_] := -4 Λ Cos[Λ] - 4 Sin[Λ] + Λ^2 Sin[Λ]
```

In[30]:=
```
Plot[FcnΛ[Λ], {Λ, 0, 10}]
```

Out[30]=


I use this visualization to calculate the first three roots and then implement a scheme that uses information about previous roots to calculate the next root.

In[31]:=
```
w[10] = With[{guesses = {2, 4, 7}},
  FindRoot[FcnΛ[x] == 0, {x, #}][[1, 2]] & /@ guesses]
```

Out[31]=
```
{1.72067, 4.05752, 6.85124}
```

In[32]:=
```
Clear[CalcRoots];
CalcRoots[roots_, nMax_] :=
 Module[{rList = Transpose[{Range[Length[roots]], roots}],
   n = Length[roots], newRoot, Guess},
  While[n < nMax,
   n = n + 1;
   (* extrapolate quadratic fit of last three roots*)
   Guess[i_] := Fit[rList[[-3 ;; -1]], {1, x, x^2}, x] /. x → n;
   newRoot = FindRoot[FcnΛ[x] == 0, {x, Guess[n]}][[1, 2]];
   AppendTo[rList, {n, newRoot}]];
  rList]
```

In[34]:=
```
w[11] = With[{nMax = 10}, CalcRoots[w[10], nMax]]
```

Out[34]=
```
{{1, 1.72067}, {2, 4.05752}, {3, 6.85124}, {4, 9.82636}, {5, 12.8746},
 {6, 15.9573}, {7, 19.0587}, {8, 22.1711}, {9, 25.2906}, {10, 28.4149}}
```

For future convenience, I create a data structure that encapsulates the eigenvalues.

In[35]:=
```
(fΛ[#[[1]]] = #[[2]]) & /@ w[11]
```

Out[35]=
```
{1.72067, 4.05752, 6.85124, 9.82636,
 12.8746, 15.9573, 19.0587, 22.1711, 25.2906, 28.4149}
```

For example,

In[36]:=
```
fΛ[7]
```

Out[36]=
```
19.0587
```

For this choice of parameters, the eigenfunctions are

In[38]:=
```
w[12] =
 w[5] /. Sol[def[Λ], λ] /. Sol[def[b], b] /. Sol[def[a], a] // PowerExpand // Quiet
```

Out[38]=

$$\psi_n[x] \to \frac{1}{2} \Lambda \,\mathrm{Cos}\left[\frac{x\,\Lambda}{L}\right] + \mathrm{Sin}\left[\frac{x\,\Lambda}{L}\right]$$

Construct

In[39]:=
```
Clear[fψ];
fψ[x_, Λ_, L_] := 1/2 Λ Cos[x Λ / L] + Sin[x Λ / L]
```

In[41]:=

```
Module[{χ = 1, L = 1, gList},
 gList = Table[ShowEigenFunction[i, L], {i, 1, 6}];
 Grid[Partition[gList, 3]]]
```

Out[41]=



In[3]:=

```
Clear[ShowEigenFunction];
ShowEigenFunction[n_, L_] :=
 Module[{Λ, lab},
  Λ = fΛ[n];
  lab = Stl@StringForm["n = `` λn = ``", n, NF3@Λ];
  Plot[1/2 Λ Cos[x Λ/L] + Sin[x Λ/L], {x, 0, L},
   PlotLabel → lab, AxesLabel → {Stl["x"], Stl["ψn[x]"]}]]
```

The next step is to use the initial condition to calculate the coefficients A_n

In[42]:=

```
w[13] = w[7] /. w[6] /. t → 0
```

Out[42]=

$$T[x, 0] == \sum_{n=1}^{\infty} A_n \psi_n[x]$$

The eigenfunctions are orthogonal, so

$$A_n = \frac{\int_0^L T(x, 0)\, \psi_n(x)\, dx}{\int_0^L \psi_n(x)\, \psi_n(x)\, dx}$$

Recall the initial condition

In[43]:=

```
A1["ic"]
```

Out[43]=

$$T[x, 0] == 1 + \frac{a x}{\kappa} - \frac{x^2 (a b L + a \kappa + b \kappa)}{L \kappa (b L + 2 \kappa)}$$

The integrals can be performed but the result is messy.

In[44]:=
```
w[14] = Integrate[(1 + a x/κ - x^2 (a b L + a κ + b κ)/(L κ (b L + 2 κ))) (1/2 Λ Cos[x Λ/L] + Sin[x Λ/L]), {x, 0, L}] /
  Integrate[(1/2 Λ Cos[x Λ/L] + Sin[x Λ/L])^2, {x, 0, L}]
```

Out[44]=
$$\left(4\left(4L\left(abL+(a+b)\kappa\right)-(aL-2\kappa)(bL+2\kappa)\Lambda^2-\left(4L\left(abL+(a+b)\kappa\right)+(aL+2\kappa)(bL+2\kappa)\Lambda^2\right)\mathrm{Cos}[\Lambda]+\kappa\Lambda\left(-2bL+2\kappa\Lambda^2+aL\left(2+\Lambda^2\right)\right)\mathrm{Sin}[\Lambda]\right)\right)\Big/\left(\kappa(bL+2\kappa)\Lambda^2\left(\Lambda\left(6+\Lambda^2\right)-2\Lambda\,\mathrm{Cos}[2\Lambda]+\left(-4+\Lambda^2\right)\mathrm{Cos}[\Lambda]\,\mathrm{Sin}[\Lambda]\right)\right)$$

In[45]:=
```
Clear[An];
An[Λ_, L_, a_, b_, κ_] :=
 (4 (4 L (a b L + (a + b) κ) - (a L - 2 κ) (b L + 2 κ) Λ^2 - (4 L (a b L + (a + b) κ) +
        (a L + 2 κ) (b L + 2 κ) Λ^2) Cos[Λ] + κ Λ (-2 b L + 2 κ Λ^2 + a L (2 + Λ^2)) Sin[Λ])) /
  (κ (b L + 2 κ) Λ^2 (Λ (6 + Λ^2) - 2 Λ Cos[2 Λ] + (-4 + Λ^2) Cos[Λ] Sin[Λ]))
```

In[47]:=
```
Clear[Tinit];
Tinit[x_, L_, a_, b_, κ_] := 1 + a x/κ - x^2 (a b L + a κ + b κ)/(L κ (b L + 2 κ));
```

In[49]:=
```
Module[{L = 1, a = 1, b = 1, κ = 1, nMax = 10, lab},
 lab = Stl@StringForm["`` term approximation of initial condition", nMax];
 Plot[
  {Tinit[x, L, a, b, κ], Sum[An[fΛ[n], L, a, b, κ] fψ[x, fΛ[n], L], {n, 1, 10}]},
  {x, 0, L}, PlotStyle → {Directive[Black, Thick], Blue},
  PlotLabel → lab, AxesLabel → {Stl["x"], Stl["T(x, 0)"]}]]
```

Out[49]=


In[50]:=
```
Clear[TSoln];
TSoln[x_, t_, L_, a_, b_, κ_, nMax_] :=
 Sum[An[fΛ[n], L, a, b, κ] e^(-t fΛ[n]) fψ[x, fΛ[n], L], {n, 1, nMax}]
```

The mixed boundary conditions imply heat flow through the boundaries and the system cools over time.

In[52]:=
```
Module[{L = 1, a = 1, b = 1, κ = 1, nMax = 10, frames},
 frames = MakeFrame[#, L, a, b, κ, nMax] & /@ Range[0, 2, 0.1];
 ListAnimate[frames]]
```

Out[52]=



In[5]:=
```
Clear[MakeFrame];
MakeFrame[t_, L_, a_, b_, κ_, nMax_] :=
 Module[{lab},
  lab = Stl@StringForm["t = ``", NF2@t];
  Plot[{TSoln[x, t, L, a, b, κ, nMax], Tinit[x, L, a, b, κ]},
   {x, 0, L}, PlotLabel → lab, AxesLabel → {Stl["x"], Stl["T(x,t)"]},
   PlotStyle → Thick, ColorFunction → Function[{x, y}, ColorData["Rainbow"][y]],
   PlotRange → {{0, L}, {0, 2}}]]
```

To check the separation of variables result, solve the same problem numerically

In[53]:=
```
SOLNNUMERICAL = Module[{L = 1, a = 1, b = 1, κ = 1, χ = 1, nMax = 10, eqns},
   NDSolve[{T^(0,1)[x, t] - χ T^(2,0)[x, t] == 0,
     -a T[0, t] + κ T^(1,0)[0, t] == 0, b T[L, t] + κ T^(1,0)[L, t] == 0,
     T[x, 0] == 1 + (a x)/κ - (x^2 (a b L + a κ + b κ))/(L κ (b L + 2 κ))}, T, {x, 0, L}, {t, 0, 1}]][[1, 1]]
```

$$T^{(0,1)}[x,t] - \chi T^{(2,0)}[x,t] = 0,\quad -a\,T[0,t] + \kappa T^{(1,0)}[0,t] = 0,\quad b\,T[L,t] + \kappa T^{(1,0)}[L,t] = 0,\quad T[x,0] = 1 + \frac{a x}{\kappa} - \frac{x^2 (a b L + a \kappa + b \kappa)}{L \kappa (b L + 2\kappa)}$$

Out[53]=
```
T → InterpolatingFunction[ Domain: {{0., 1.}, {0., 1.}} Output: scalar ]
```

In[55]:=

```
Plot3D[T[x, t] /. SOLNNUMERICAL, {x, 0, 1},
 {t, 0, 1}, ColorFunction → "TemperatureMap",
 AxesLabel → {Stl["x"], Stl["t"], Stl["T(x,t)"]}, PlotLabel → "Numerical solution",
 Mesh → False, Boxed → False, PlotLegends → Automatic]
```

Out[55]=



Compare the separation of variable solution with the numerical solution at x = L/2

In[56]:=

```
Module[{L = 1, a = 1, b = 1, ϰ = 1, χ = 1, nMax = 10, numerical},
 numerical = Table[{t, (T[L/2, t] /. SOLNNUMERICAL)}, {t, 0, 1, 0.1}];
 Plot[TSoln[L/2, t, L, a, b, ϰ, nMax], {t, 0, 1}, PlotRange → All,
  Epilog → {OC[#, Black] & /@ numerical}, AxesLabel → {Stl["t"], Stl["T(L/2, t)"]},
  PlotLabel → Stl["Comparison of analytical solution (line)\n
with numerical solution (circles)"], PlotRange → All]]
```

Out[56]=



A Construction of initial conditions consistent with boundary conditions

The boundary conditions are

In[57]:=
```
wA[1] = {κ T^(1,0)[0, t] - a T[0, t] == 0, κ T^(1,0)[L, t] + b T[L, t] == 0}
```

Out[57]= $\{-a\,T[0,t]+\kappa\,T^{(1,0)}[0,t]=0,\ b\,T[L,t]+\kappa\,T^{(1,0)}[L,t]=0\}$

I construct a function f(x) that satisfies both of these conditions

In[58]:=
```
wA[2] = wA[1] /. T → Function[{x, t}, f[x]]
```

Out[58]= $\{-a\,f[0]+\kappa\,f'[0]=0,\ b\,f[L]+\kappa\,f'[L]=0\}$

Assume the initial temperature profile is quadratic and choose the coefficients to satisfy the boundary conditions.

In[59]:=
```
wA[3] = wA[2] /. f → Function[{x}, α x^2 + β x + γ]
```

Out[59]= $\{-a\,\gamma+\beta\,\kappa=0,\ b\,(L^2\alpha+L\,\beta+\gamma)+(2\,L\,\alpha+\beta)\,\kappa=0\}$

In[60]:=
```
wA[4] = Solve[wA[3], {α, β}][[1]]
```

Out[60]= $\left\{\alpha\to-\frac{\gamma\,(a\,b\,L+a\,\kappa+b\,\kappa)}{L\,\kappa\,(b\,L+2\,\kappa)},\ \beta\to\frac{a\,\gamma}{\kappa}\right\}$

In[61]:=
```
wA[5] = α x^2 + β x + γ /. wA[4]
```

Out[61]= $\gamma+\frac{a\,x\,\gamma}{\kappa}-\frac{x^2\,\gamma\,(a\,b\,L+a\,\kappa+b\,\kappa)}{L\,\kappa\,(b\,L+2\,\kappa)}$

In[62]:=
```
wA[6] = wA[5] /. γ → 1
```

Out[62]= $1+\frac{a\,x}{\kappa}-\frac{x^2\,(a\,b\,L+a\,\kappa+b\,\kappa)}{L\,\kappa\,(b\,L+2\,\kappa)}$

For example

In[63]:=

```
Module[{L = 1, κ = 1, a = 1, b = 1},
 Plot[1 + a x/κ - x^2 (a b L + a κ + b κ)/(L κ (b L + 2 κ)), {x, 0, L}]]
```

$$1+\frac{a\,x}{\kappa}-\frac{x^2\,(a\,b\,L+a\,\kappa+b\,\kappa)}{L\,\kappa\,(b\,L+2\,\kappa)}$$

Out[63]=



Functions

In[7]:=

```
Clear[ShowPDESetup];
ShowPDESetup[A_] := Module[{top = 1.0, right = 1.0,
    boundaries, labels, textInterior, textIC, textBCL, textBCR},
   boundaries = Line /@ {{{0, 0}, {right, 0}},
      {{0, 0}, {0, top}}, {{right, 0}, {right, top}}};
   labels = Text[PhysicsForm[A[#[[1]]]], #[[2]]] & /@
     {{"pde", {right/2, top/2}}, {"ic", {right/2, 0.0}},
      {"bcL", {0.0, top/2}}, {"bcR", {right, top/2}}};
   Graphics[{Directive[Black, Thick], boundaries, labels}, Axes → False,
    AspectRatio → 0.25, ImageSize → 500, PlotLabel → Stl[A["description"]]]]
```

In[8]:=

```
Clear[DSolveHeatEquation];
DSolveHeatEquation[A_] :=
 Module[{soln},
  soln = DSolve[A["eqns"], T[x, t], {x, t}, Assumptions → A["assumptions"]][[1, 1]];
  soln = soln //. A["substitutions"];
  soln = Simplify[soln, A["simplifications"]];
  soln]
```